

Mapping properties of exponential function (continued)

In this note, we will use Mathematica to plot the image of a region (rectangle, square, triangle, disk, etc) under the exponential map $f(z) = e^z$.

If we know a parametrization of a region, either in real form

$$\begin{cases} x = g(t, s) \\ y = h(t, s) \end{cases} \quad a \leq t \leq b, \quad c \leq s \leq d$$

or in complex form

$$z = g(t, s) + ih(t, s),$$

we can plot the region by the command **ParametricPlot**. This command has several different syntaxes, one of which is:

```
ParametricPlot[{g[t,s],h[t,s]}, {t, a, b}, {s, c, d}]
```

or equivalently

```
ParametricPlot[ReIm[g[t,s] + I*h[t,s]], {t, a, b}, {s, c, d}]
```

where $g(x, y)$ and $h(x, y)$ are two scalar-valued functions. The above command draws the set of all points of coordinate $(g(x, y), h(x, y))$ where (x, y) runs over the rectangle $[a, b] \times [c, d]$. For example (Figure 1),

```
ParametricPlot[{t*Sin[s], s*Sin[t]}, {t, -2, 2}, {s, -2, 2}]
```




Figure 1:

The range of t or s can involve a variable. For example, the right triangle with vertices at $(0, 0)$, $(0, 2)$, $(1, 0)$ has parametrization $x = t$, $y = s$ where $0 \leq t \leq 1$ and $0 \leq s \leq 2 - 2t$. In Mathematica (Figure 2),

```
ParametricPlot[{t, s}, {t, 0, 1}, {s, 0, 2 - 2*t}]
```

To plot the image of this triangle under the exponential function, we write (Figure 3)

```
f[z_] := Exp[z]
ParametricPlot[ReIm[f[t+s*I]], {t, 0, 1}, {s, 0, 2 - 2*t}]
```

We know that the $f(z) = e^z$ is periodic with period $2\pi i$. To see all possible values of $f(z)$ (or in other words, the range of f), it suffices to restrict the domain of f to an infinite horizontal strip of width 2π , for instance $(-\infty, \infty) \times (-\pi, \pi]$, instead of the whole complex plane. To view the image of the rectangle $[-1, 1] \times [-\pi, \pi]$ under f , use the **ParametricPlot** command. To draw the image of a rectangle $[-1, 1] \times [-\pi, \pi]$ under the function e^z , we write (Figure 4)




Figure 2:




Figure 3:

```
ParametricPlot[ReIm[f[x+I*y]], {x, -1, 1}, {y, -Pi, Pi},
               PlotRange -> Full]
```

To see how each grid line $x = a$ and $y = b$ is transformed into, one simply adds the option **Mesh** and **MeshStyle** to the above command (Figure 5).

```
ParametricPlot[ReIm[f[x + I*y]], {x, -1, 1}, {y, -Pi, Pi},
               PlotRange -> Full, Mesh -> Automatic, MeshStyle -> {Red, Green}]
```

The grid lines $x = a$ are transformed into red curves (circles), while the grid lines $y = b$ are transformed into green curves (straight segments emitting from the origin). It would be helpful to see those transformations in motion. To draw the rectangles $[-s, s] \times [-\pi, \pi]$ side by side with their images as s runs from 1 to 2, use the commands (Figure 6):

```
p[s_] := ParametricPlot[ReIm[x + y*I], {x, -s, s}, {y, -Pi, Pi},
                       PlotRange -> {{-2, 2}, {-Pi, Pi}}, MeshStyle -> {Red, Green},
                       Mesh -> Automatic]

q[s_] := ParametricPlot[ReIm[f[x + y*I]], {x, -s, s}, {y, -Pi, Pi},
                       PlotRange -> {{-8, 8}, {-8, 8}}, MeshStyle -> {Red, Green},
                       Mesh -> Automatic]

Manipulate[{p[s], q[s]}, {s, 1, 2}]
```

Moving the cursor to vary s , one can see that the image of each rectangle is an annulus. The outer radius grows larger while the inner radius shrinks as $s \rightarrow \infty$. A good guess is that the image of the infinite strip $(-\infty, \infty) \times (-\pi, \pi]$ under the exponential function is $\mathbb{C} \setminus \{0\}$, i.e. the whole complex plane except for the origin (a punctured plane).

Let us look at Figure 6. The **ParametricPlot** command not only shows the image the rectangle on the left, but also tells us if a point on the right graph comes from multiple points on the left graph. The light blue background on the right indicates that each point on the right comes from exactly one point on the left. Let us choose the range of y to be $[-\pi, s]$ and then let s vary from $\pi - 1$ to $\pi + 1$:




Figure 4:




Figure 5:

```
p[s_] := ParametricPlot[ReIm[x + y*I], {x, -1, 1}, {y, -Pi, s},
  PlotRange -> {{-1, 1}, {-Pi, Pi + 1}}, MeshStyle -> {Red, Green},
  Mesh -> Automatic]

q[s_] := ParametricPlot[ReIm[f[x + y*I]], {x, -1, 1}, {y, -Pi, s},
  PlotRange -> {{-3, 3}, {-3, 3}}, MeshStyle -> {Red, Green},
  Mesh -> Automatic]

Manipulate[{p[s], q[s]}, {s, Pi - 1, Pi + 1}]
```

When $s < \pi$, the image under f of the rectangle $[-1, 1] \times [-\pi, s]$ is not yet a full annulus (Figure 7). When $s > \pi$, there occurs a darker blue region on the right graph (Figure 8), suggesting that each point in this region comes from two points on the left. Thus, it is reasonable to conclude that the exponential function maps

- the strip $(-\infty, \infty) \times (-\pi, \pi]$ one-to-one and onto (in another word, bijectively) the punctured complex plane $\mathbb{C} \setminus \{0\}$,
- the open strip $(-\infty, \infty) \times (-\pi, \pi)$ one-to-one and onto the slit complex plane $\mathbb{C} \setminus \mathbb{R}_{\leq 0}$.

A careful calculation can confirm these conclusions. However, detail computations are somewhat complicated and will be skipped.




Figure 6:




Figure 7:




Figure 8: